

Azterketa Zati praktikoa

Topografia berezitasuna
Informatika I
Data: 2006-6-23

1. [2,5 puntu] **Analizatu, diseinatu eta idatzi** programa bat, neurketa topografiko batean lortutako eta ibilbide bat osatzen duten puntu sekuentzia finitu baten altuerak irakurri eta puntu bakoitza kolore baten arabera sailkatzeko mezua pantailaratzen duena. Horretarako, galdetu erabiltzaileari zenbat puntu neurtu dituen. Erabilitako koloreak: berde argia 200m baino baxuagoa bada, berdea 200m eta 600m artean badago eta berde iluna 600 baino altuagoa izan ezker.

Oharra: diseinua fluxu-diagrama bidez adierazi eta programa VisualBasic lengoia erabiliz.


Analisia:

- Sarrera:
 1. Puntu sekuentziaren kopurua (kopurua)
 2. Zenbaki oso positiboen sekuentzia bat (altuera)
- Irteera: Koloreen sekuentzia bat, emandako puntu sekuentziaren altuerak sailkatzen dituen (kolorea)
- Prozesuak:
 1. Errepikatu sekuentziako puntu bakoitzeko (kont)
Altueraren irakurketa
2. Altueraren sailkapena
 2. Altueraren sailkapena:
Kasu1: $\text{altuera} < 200 \rightarrow \text{kolorea} = \text{"berde argia"}$
Kasu2: $200 \leq \text{altuera} \leq 600 \rightarrow \text{kolorea} = \text{"berdea"}$
Kasu3: $\text{altuera} > 600 \rightarrow \text{kolorea} = \text{"berde iluna"}$

Azterketa Zati praktikoa

Topografia berezitasuna
 Informatika I
 Data: 2006-6-23

Diseinua:



Azterketa Zati praktikoa

Topografia berezitasuna
Informatika I
Data: 2006-6-23

Programa:

```
Public Sub Ariketa1_Click()  
    Dim kopurua As Integer, altuera As Integer, kont As Integer  
    Dim kolorea As String  
  
    Kopurua = InputBox("Eman puntu sekuentziaren luzera")  
    For kont = 1 To kopurua  
        Altuera = InputBox(kont & ". Puntuaren altuera eman:")  
        If Altuera < 200 Then  
            Kolorea = "berde argia"  
        Else If Altuera > 600 Then  
            Kolorea = "berde iluna"  
        Else  
            Kolorea = "berdea"  
        End If  
        End If  
        MsgBox(kont & ". puntuaren kolorea: " & kolorea)  
    Next kont  
End Sub
```

2. [3 puntu] MENDIZALE mendi elkarteak ondorengo interfazea edukiko duen programa bat garatu nahi du bere mendi irteerak gestionatzeko:



Mendi elkarteak 20 mendi irteera antolatzen ditu urtean. Mendi irteera bakoitzerako ondorengo informazioa gordetzen delarik: mendiaren izena (Karaktere-katea), irteera data (8 digituko zenbaki osoa, uuuuhhee formatuarekin), altuera (Komaduna) eta denbora (Komaduna).

Jakinda programak informazioa irteera dataren arabera **ordenatuta** duela, balio txikienetik handienera, ondorengo atalak egitea eskatzen da (ez programa osoa):

- 1.- Definitu informazioa gordetzeko behar dituzun datu-egitura komunak.
- 2.- Ondorengo eragiketak, bakoitza bere aldetik, ebazteko behar den kodea idatzi:
 - a.- Mendirik altuena botoia sakatzean, **mendi altuena bilatu eta** mendi horri dagozkion datuak pantailaratu aldameneko irudi-kutxan.
 - b.- Hurrengo mendi irteera botoia sakatzean, aurrez testu-kutxan erabiltzaileak sartutako data horretatik **aurrera** hurbilen dagoen hurrengo mendi irteeraren datuak (gehienez bakar batenak) pantailaratuko dira, aldameneko irudi-kutxan. **(Suposatu egun batean mendi irteera bakar bat dagoela eta testu-kutxan sartutako balioa uuuuhhee formatua betetzen duen zenbaki oso bat dela)**

Oharra: Testu-kutxetan sartu diren datuak zuzenak direla suposatzen da.

Visual Basic-eko Objektuen laburpena:


Kontrol-objektua	Mota	Propietatea	Balioa
cmdSartu	CommandButton (cmd)	Name	cmdSartu
		Caption	Datuak sartu
cmdAltuena	CommandButton (cmd)	Name	cmdAltuena
		Caption	Mendi altuena
txtData	TextBox (txt)	Name	txtData
		Text	
cmdBilatu	CommandButton (cmd)	Name	cmdBilatu
		Caption	Hurrengo mendi irteera
pctEmitza	PictureBox (pct)	Name	pctEmitza

Azterketa Zati praktikoa

Topografia berezitasuna
Informatika I
Data: 2006-6-23

1. Definitu informazioa gordetzeko behar dituzun datu-egitura komunak

Grafikoki:



Visual Basic kodeaz:

```
' Mendi irteera baten informazioa gordetzeko datu-egitura (hots, erregistroa)
Private Type IrteeraMota
    Izena As String
    Data As Long ' Suposatuz data adierazteko formatua uuuuhhee dela
    Altuera As Single
    Denbora As Single
End Type

' 20 mendi irteera gordetzeko datu-egitura baten definizioa (hots, taula)
Const IRTEERA_KOP As Integer = 20
Dim Irteerak(1 To IRTEERA_KOP) As IrteeraMota
```

2.a-

```
Private Sub cmdAltuena_Click()
    Dim altuena As Single ' Mendi altuenaren altuera
    Dim indAltuena As Integer ' Mendi altuenaren informazioa taulan duen posizioa
    Dim indizea As Integer ' Aztertzen ari garen mendiak taulan duen posizioa

    altuena = Irteerak(1).Altuera ' Altuena lehenengoa dela suposatzen da
    indAltuena = 1
    For indizea = 2 To IRTEERA_KOP ' mendi irteera guztiak aztertu
        If Irteerak(indizea).Altuera > altuena Then
            altuena = Irteerak(indizea).Altuera
            indAltuena = indizea
        End If
    Next indizea
    pctEmaita.Print("Mendirik altuena: " & vbCrLf & _
        vbCrLf & "Izena: " & Irteerak(indAltuena).Izena & vbCrLf & _
        vbCrLf & "Data: " & Irteerak(indAltuena).Data & vbCrLf & _
        vbCrLf & "Altuera: " & Irteerak(indAltuena).Altuera & vbCrLf & _
        vbCrLf & "Denbora: " & Irteerak(indAltuena).Denbora )
End Sub
```

Azterketa Zati praktikoa

Topografia berezitasuna
Informatika I
Data: 2006-6-23

2.b-

```
Private Sub cmdBilatu_Click()  
    Dim data As Long      ' uuuuhhee formatua duen zenbaki osoa  
    Dim indizea As Integer ' aztertzen ari garen mendi irteeraren posizioa taulan  
    Dim aurkitua As Boolean ' ez direla aurkitu mezua eman nahi izan ezker eta  
                                ' taulako datu guztiak ez aztertzeko  
  
    ' Sarrerako datuak irakurri eta aldagaiak hasieratu  
    data = Val(txtData.Text)  
    aurkitua = False  
    indizea = 1  
  
    ' Bitartean mendi irteerak daude aztertzeko eta ez da hurrengoa aurkitu  
    While indizea <= IRTEERA_KOP And Not (aurkitua)  
        ' emandako data horretatik aurrera hurbilen dagoen hurrengo irteera zein den aztertu  
        If Irteerak(indizea).data >= data Then  
            aurkitua = True ' gutxienez bat badago  
        Else  
            indizea = indizea + 1  
        End If  
    Wend  
  
    ' Emaita idatzi  
    If aurkitua Then  
        pctEmaita.Print (" Hurrengo mendi irteera: " & vbCrLf & _  
            vbCrLf & "Izena: " & Irteerak(indizea).Izena & vbCrLf & _  
            vbCrLf & "Data: " & Irteerak(indizea).Data & vbCrLf & _  
            vbCrLf & "Altuera: " & Irteerak(indizea).Altuera & vbCrLf & _  
            vbCrLf & "Denbora: " & Irteerak(indizea).Denbora)  
    Else  
        pctEmaita.Print ("Ez dago mendi irteerarik " & data & " data horretatik aurrera")  
    End If  
End Sub
```

Azterketa Zati praktikoa

Topografia berezitasuna
Informatika I
Data: 2006-6-23

3. [2 puntu] **Espezifikatu eta idatzi** azpiprograma bat, non esaldi bat eta bi karaktere emanda, esaldian agertzen diren lehen **karakterearen** agerpen guztiak 2. **karakterearekin** trukatzeko dituen.

Adibidez: esaldia: "Karaktereak trukatu egingo dira."

1go karakterea: " " eta 2. karakterea " ," izanik

Irteera: Karaktereak,trukatu,egin,dira.

Bi aukera daude Funtzio edo Prozedura bat definitzea. **Funtzioa** definitzea askoz egokiagoa da, ondorengo arrazoiengatik: **1) Balio bat bakarrik bueltatzen du; 2) ez da sarrera-irteera agindurik egitea eskatzen eta 3) adierazpenetan erabiltzeko aukera ematen du.**

```
Public Function Trukatu(ByVal esaldia As String, ByVal Kar1 As String, _  
    ByVal Kar2 As String) As String  
    'Aurrebaldintza: Kar1 eta Kar2 parametroek karaktere bakar bat adierazten dute  
    'Postbaldintza: Itzultzen den karaktere katea esaldia parametroak duen balioaren ia berdina  
    'da, hots, karaktere guztiak posizio berdinean daude Kar1 izan ezik, honen  
    'agerpen guztietan Kar2 agertzen da  
  
    Dim esaldiBerria As String ' Emaizka kalkulatzeko  
    Dim pos As Integer        ' Aztertu beharreko karakterearen posizioa  
    Dim kar As String         ' Aztertu beharreko karakterea  
    Dim luzera As Integer  
  
    ' Trukaketako karaktereak berdinak badira ez da ezer egin behar  
    If Kar1 = Kar2 Then  
        esaldiBerria = esaldia  
    Else  
        esaldiBerria = ""  
        'Esaldi berria karakterez-katere sortzen joan, Kar1 agertzen den bakoitzean  
        'Kar2-rekin ordezkatzuz  
        luzera = Len(esaldia)  
        For pos = 1 To luzera  
            kar = Mid(esaldia, pos, 1)  
            If kar = Kar1 Then  
                kar = Kar2  
            End If  
            esaldiBerria = esaldiBerria & kar  
        Next pos  
    End If  
    'Emaizka itzuli  
    Trukatu = esaldiBerria  
End Function
```

Azterketa Zati praktikoa

Topografia berezitasuna
Informatika I
Data: 2006-6-23

Beste aukera bat:

```
Public Function Trukatu(ByVal esaldia As String, ByVal Kar1 As String, _  
    ByVal Kar2 As String) As String  
    'Aurrebaldintza: Kar1 eta Kar2 parametroek karaktere bakar bat adierazten dute  
    'Postbaldintza: Itzultzen den karaktere katea esaldia parametroak duen balioaren ia berdina  
    '    da, hots, karaktere guztiak posizio berdinean daude Kar1 izan ezik, honen  
    '    agerpen guztietan Kar2 agertzen da  
  
    Dim esaldiBerria As String 'Emaita kalkulatzeko  
    Dim pos As Integer        ' Kar1 agertzen den posizioa adierazten du  
    Dim azpikatea As String  ' Kopiatu beharreko azpikatea  
    Dim hasiPos As Integer    ' karaktereak kopiatzen hasteko posizioa adierazten du  
    Dim kopurua As Integer    ' Zenbat karaktere kopiatu adierazten du  
    Dim luzera As Integer  
  
    ' Trukaketako karaktereak berdinak badira ez da ezer egin behar  
    If Kar1 = Kar2 Then  
        esaldiBerria = esaldia  
    Else  
        ' hasieraketak  
        esaldiBerria = ""  
        hasiPos = 1  
        pos = InStr(hasiPos, esaldia, Kar1)  
  
        ' Kar1 agertzen den bakoitzeko, aurreko zatia kopiatu eta Kar2 gehitu  
        While pos > 0  
            kopurua = pos - hasiPos  
            azpikatea = Mid(esaldia, hasiPos, kopurua)  
            esaldiBerria = esaldiBerria & azpikatea & Kar2  
            hasiPos = pos + 1  
            pos = InStr(hasiPos, esaldia, Kar1)  
        Wend  
  
        ' Esaldiaren bukaeran karaktereak gelditzen al dira kopiatzeko?  
        luzera = Len(esaldia)  
        If hasiPos <= luzera Then  
            kopurua = luzera - hasiPos + 1  
            esaldiBerria = esaldiBerria & Right(esaldia, kopurua)  
        End If  
    End If  
    'Emaita itzuli  
    Trukatu = esaldiBerria  
End Function
```


Azterketa Zati praktikoa

Topografia berezitasuna
Informatika I
Data: 2006-6-23

4. [1,5 puntu] Idatzi programa bat, erabiltzaileari zenbaki bitar bat eskatzen diona eta bere osagarria kalkulatu eta pantailaratzen duena. Programaren soluzioan azpiprogramak erabili.

Adibidez: Eman zenbaki bitar bat: **00111111**

Bere osagarria: **11000000** da.

Oharra: Zenbaki bitar bat 0 eta 1 karaktereez osatutako zenbaki bat da. Erabiltzaileak emandako balioak hori betetzen duela ziurtatu, hala ez den kasuan dagokion errore-mezua emanaz. **Ondorengo azpiprograma ere erabili daiteke, ez da inplementatu beharrik.**

Public Function ZenbakiBitarraDa(ByVal zenbakiBitarra As String) As Boolean

'Aurrebaldintza:

'Postbaldintza: *True* balioa bueltatzen du, *zenbakiBitarra* parametroa 0 eta 1

' karaktereez osatuta badago eta *False*, bestelako kasuan.

Private Sub Ariketa4 ()

Dim zenbakia As String ' Erabiltzaileak emandako zenbakia

Dim partziala As String ' Emaizta partziala gordetzeko

Dim osagarria As String ' Azken emaitza

zenbakia = InputBox("Eman zenbaki bitar bat: ")

If zenbakiBitarraDa(zenbakia) Then

 ' karaktere laguntzaile bat behar da bi karaktereen arteko trukaketa egiteko

 partziala = Trukatu(zenbakia, "0", "z")

 partziala = Trukatu(partziala, "1", "0")

 osagarria = Trukatu(partziala, "z", "1")

 MsgBox ("Bere osagarria: " & osagarria)

Else

 MsgBox ("Emandako zenbakia ez da zenbaki bitarra")

End If

End Sub